Dynamische Toolbars in Aktion

Lisa Slater Nicholls 

In dem Artikel „Dynamische Toolbars für dynamische Produktivität" haben wir eine Toolbar-Klasse für eine Entwicklungsumgebung erstellt. Dieses Mal behandeln wir den Extended Database Container (EDC), ein Public Domain-Tool von Tom Rettig, das Visual FoxPros Database Container (DBC) erweitert. Dann leiten wir von unserer Entwicklungs-Toolbar eine Klasse ab, die uns einen einfachen Zugriff auf den EDC ermöglicht, während wir unsere Anwendung erstellen. 


Einführung in den EDC 

Der EDC ist eine der Klassenbibliotheken in TRUE (Tom Rettig's Utility Extensions for Visual FoxPro). TRUE ist eine Anzahl von Public Domain-Tools, entwickelt von der Tom Rettig Micro Corp. 

TRUE ist ein Beispiel für die Funktionalität und das elegante Design, die Sie in TRO (Tom Rettig's Office) finden, einer Sammlung von Tools für die schnelle Entwicklung einer Anwendung, die in einer Geschäftsanwendung integriert sind. Jede Klassenbibliothek in TRUE allein ist schon sehr hilfreich und kann in jede Problemlösung, die Sie entwickeln, integriert werden. Das möchte ich Ihnen anhand des EDC demonstrieren. 

TRUE.EXE enthält prozedurbasierende Versionen der einzelnen Klassenbibliotheken. Zum Beispiel ist der EDC in TRUE.EXE als EDCLIB.PRG enthalten. TRUE.EXE enthält auch das Programm PRGTOVCX.PRG, das jede prozedurbasierende Bibliothek in eine visuelle Bibliothek umwandelt. Wenn Sie die Beispieldateien dieses Artikels benutzen, finden Sie dort auch EDCLIB.VCX, die visuelle Klassenbibliothek, die mit Hilfe von PRGTOVCX.PRG erstellt wurde. 

Was erweitert der EDC? 

Wie Sie wahrscheinlich wissen, bestehen die Datenbankcontainer von Visual FoxPro aus Tabellen, deren Felder alle Objekte, die zu den Containern gehören (zum Beispiel Relationen, Felder oder Tabellen), beschreiben. Wenn Sie Informationen über die Tabellen in einer Datenbank benötigen, können Sie die Funktion DBGETPROP() aufrufen, um den DBC abzufragen. Als Beispiel fragt der folgende Code den DBC nach dem Primärschlüssel der Tabelle, die in der aktuellen Datenbank (TEST.DBC) enthalten ist: 

OPEN DATA Test 
SET DATA TO Test 
? DBGETPROP("Table1", "TABLE", "PRIMARYKEY") 

Beachten Sie, daß DBGETPROP() drei Parameter benötigt: den Namen des Objekts (Tabelle1 in diesem Beispiel), den Typen des Objekts (Table) und die Eigenschaft, die Sie erfahren möchten (der Primärschlüssel der Tabelle). 

Jeder Typ der Objekte, die in einem DBC enthalten sind, hat seine eigenen Eigenschaften. Wenn Sie den DBC nach einem Feld (statt nach einer Tabelle) abfragen, können Sie nicht nach einem Primärschlüssel fragen; das würde keinen Sinn machen. Statt dessen könnten Sie nach der Caption-Eigenschaft oder einem Default-Wert fragen. Nachdem Sie die Informationen aus dem DBC abgefragt haben, können Sie vielen Eigenschaften mit Hilfe der Funktion DBSETPROP() einen neuen Wert zuweisen. Die Syntax dieser beiden Funktionen ist ähnlich. 

Warum den DBC erweitern? 

Werfen Sie einmal einen Blick auf die Eigenschaften jedes Objekttyps, der im DBC enthalten ist. Sie werden feststellen, daß einige wichtige Informationen fehlen. Zum Beispiel sind zum Feld nicht die dazugehörigen Eingabemasken gespeichert. Andersherum können Sie, wenn Sie den DBC allein benutzen, diese Information nicht Eingabemasken zuordnen. Mit dieser Information könnte man sicherstellen, daß bei jeder Benutzung des Feldes die gleiche Formatschablone verwendet wird. 

Formatschablonen sind mehr als nur Programmkosmetik; sie sind eine einfache Methode, die Gültigkeit von Daten bereits im Vorfeld zu prüfen. Die Formatschablone stellt sicher, daß eine Postleitzahl oder eine Telefonnummer entsprechend der Spezifikationen der Applikation formatiert ist, oder daß alle Paßwörter automatisch groß geschrieben sind. Eine Schablone stellt auch eine Prüfung der Länge der Eingabe zur Verfügung, was besonders dann sinnvoll ist, wenn man nichtproportionale Schriften für die Dateneingabe verwendet. Sie können die InputMask-Eigenschaft auch verwenden, um das Datenformat „on the fly" zu ändern. Das könnte für ein Feld Postleitzahlen sinnvoll sein, wenn Sie sich durch eine Adreßtabelle mit Datensätzen aus verschiedenen Ländern bewegen. Unabhängig davon, ob die Erfordernisse der Formatierung statisch oder dynamisch sind, eine Formatschablone benötigen Sie immer. Wenn Sie es wünschen, können Sie eine spezielle Kontrolle für die Dateneingabe für jedes Feld in jeder Tabelle Ihrer Datenbank erstellen und für alle auch Eingabemasken definieren. Sie werden schnell die Ineffektivität Ihres Vorgehens bemerken, wenn Sie durch Ihre Klassen jagen, um Ihre Formulare zu designen. Der richtige Platz, diese Daten zu speichern, ist bei den Daten _ und genau das macht der EDC. 

Dieses Beispiel stellt in VFP 3.0 Funktionalität bereit, die bereits im Basisprodukt von VFP 5.0 enthalten ist. Es gibt aber noch viele andere Datenattribute, die Sie auch im DBC von VFP 5.0 oder 6.0 nicht speichern können. Wenn Sie erst einmal verstanden haben, wie der EDC den DBC erweitert, werden Ihnen viele Gründe einfallen, den EDC zu benutzen. 

Wie EDC arbeitet 

Da DBCs aus Tabellen bestehen, die Datensätze für jedes in ihnen enthaltene Objekt enthalten, kann die EDC-Klassenbibliothek einfach eine zweite Tabelle erstellen, in der zusätzliche Informationen über jedes Objekt gespeichert werden. EDC schreibt seine Signatur in das Anwenderfeld des DBC und erstellt dann eine XBASE-Relation zum DBC. (Andere Entwicklungswerkzeuge können dem Anwenderfeld Daten hinzufügen, ohne den EDC-Link zu zerstören.) 

In einer objektorientierten Umgebung brauchen Sie nicht zu wissen, wie der EDC arbeitet. Die Bibliothek stellt Ihnen Funktionen zur Verfügung, die es Ihnen ermöglichen, den Link einzurichten, Informationen im EDC zu speichern und andere Informationen zurückzubekommen. Sie erstellen eine Instanz der EDC-Klasse und diese Funktionen sind der einzige Teil, den Sie normalerweise zu sehen bekommen. 

Als erstes müssen Sie die Tabelle mit den EDC-Erweiterungen initialisieren und ihr mitteilen, welche zusätzlichen Informationen Sie über die Objekte, die sich im EDC befinden, speichern wollen. Ich habe das Beispielprogramm KICKSTART.PRG meinen Beispieldateien hinzugefügt, um Ihnen zu zeigen, welche Kommandos Sie benutzen können, um den Eintrag für eine Formatschablone für jede der zwei Tabellen in TEST.DBC zu erstellen und abzufragen. Der folgende Code erstellt die EDC-Tabelle: 

SET LIBRARY TO EDCLIB ADDITIVE 
xx = CREATEOBJECT("edc") 
xx.lOpen("Master") 

Die Funktion lOpen() des EDC erstellt die EDC-Tabelle, da sie noch nicht existiert, und stellt die Verbindung zum aktuellen DBC her. Der „Master"-Parameter ist der Name des Feldes, das meine erweiterten Eigenschaften enthält. Eine EDC-Tabelle kann eine beliebige Anzahl dieser Felder enthalten _ vorausgesetzt, Sie wollen auf verschiedene erweiterte Eigenschaften für verschiedene Anwendungsgebiete zugreifen. Zum Beispiel könnten verschiedene Überschriftseigenschaften für verschiedene internationale Versionen Ihrer Applikation benutzt werden. Falls Sie schon einmal mit Tom Rettigs TRO gearbeitet haben, werden Sie wahrscheinlich feststellen, daß TRO dem EDC ein zweites Feld hinzufügt, um seine eigenen Informationen zu speichern, während Ihre erweiterten Eigenschaften unberührt bleiben. 

Wenn wir unser einfaches Beispiel erstellen, behalten Sie im Gedächtnis, daß wir nur einen Teil der Möglichkeiten benutzen, die uns EDC durch die Speicherung und den Gebrauch von Metadaten an die Hand gibt. Zusätzlich zu den mehrfachen Erweiterungsfeldern kann EDC auch Informationen über Tabellen speichern, die nicht Teil eines einzelnen DBC sind, und es kann auch mehr als ein EDC für die Unterstützung eines DBC genutzt werden. 

Ich empfehle Ihnen allerdings, daß Sie jeden EDC mit einem speziellen DBC verbinden und auch nur einen EDC je DBC verwenden, bis Sie mit dem Linking-Prozeß wirklich vertraut sind. Außerdem brauchen Sie wahrscheinlich am Anfang nicht mehr als ein Erweiterungsfeld. 

Lassen Sie uns auf die Initialisierung einer Formatschablone in unserer einfachen Erweiterung zurückkommen. Hier ist die Codezeile, die die EDC-Einträge erstellt, in einer FOR/ENDFOR-Schleife, die diese Zeile für jedes Feld einer Tabelle wiederholt. 

USE table1 
FOR yy = 1 to FCOUNT() 
? xx.uSetProp("table1. " + field(yy), "FIELD", ; 
"MyInputMask", " ") 
ENDFOR 

Die EDC-Funktion uSetProp(), die die Eigenschaften setzt, entspricht in der Syntax der Funktion DBSETPROP() für den eingebauten Datenbankcontainer. Sie übergeben der Funktion die folgenden Informationen: 

· Den Datensatz, den Sie editieren möchten 
hier habe ich den Datensatz mit dem Ausdruck "table1"+field(yy) referenziert) 

· Den Typ des Datensatzes (hier: FIELD) 

· Die Eigenschaft, die Sie einfügen wollen 
(hier: MyInputMask) 

· Den Wert für diese Eigenschaft 
(hier: ein Leerzeichen für jedes Feld, da wir die Input-Masken später individuell verbinden wollen) 

Was hat das mit visuellem Design zu tun? 

Ist Ihnen aufgefallen, daß ich keine Screenshots benötigt habe, um diesen Prozeß zu illustrieren? Das liegt daran, daß sich bis jetzt alles im Code abgespielt hat, weit weg von den Design-Tools. Wenn es Ihnen wie mir geht, interessiert es Sie vielleicht, zum Datenbank-Designer und seinen Dialogfeldern zurückzukehren, um mit Ihrer Entwicklungsarbeit fortzufahren. 

Moment: Die Sache hat einen Haken. Der Datenbank-Designer erhält Informationen von den Dialogboxen und stellt sie in die DBC-Tabelle. Der Datenbank-Designer weiß absolut nichts vom EDC. Dafür benötigen wir ein neues visuelles Werkzeug, das als Front-End für den EDC dient. 

Klicken Sie mit der rechten Maustaste auf den Datenbank-Designer. Sie sehen, daß der Datenbank-Designer, mit Ausnahme der referentiellen Integrität, nicht die Möglichkeit besitzt, Builder für die verschiedenen Elemente eines DBC einzusetzen. Obwohl jeder Builder, der als ALL oder RI registriert ist, erscheint, wenn Sie aus dem Right-Klick-Menü referentielle Integrität auswählen, bekommen Sie diese Möglichkeit nur, wenn Sie eine Relation ausgewählt haben. (Das ist eine Einschränkung von VFP, kein Problem der _BUILDER-Applikation.) Wir brauchen nicht nur ein neues visuelles Tool, wir benötigen einen neuen Weg, dieses Tool aufzurufen. 

Öffnen Sie die Toolbar 

Nun ist es an der Zeit, die Klasse lsnToolbar einmal laufen zu lassen, die wir in dem Artikel „Dynamische Toolbars für dynamische Produktivität" erstellt haben. Wenn Sie die Klasse aufrufen, stellt die Toolbar Dienste zur Verfügung, die in der Entwicklungsumgebung notwendig sind: intelligentes Aktivieren und Deaktivieren seiner eingebetteten Objekte, wenn es notwendig ist, mit Hilfe des Timers, mit dem sie zusammenarbeitet. In unserem Beispiel haben wir eine Klasse abgeleitet, die einem Entwickler hilft. 

Dieses Mal wollen wir von lsnToolbar die Klasse lsnToolbarEDC ableiten. Um der Toolbar EDC-Funktionalität hinzuzufügen, fügen Sie einfach die Klassenbibliothek EDCLIB.VCX der Formular-Steuerelemente-Toolbar hinzu, indem Sie auf den Button „Klassen anzeigen" klicken. Nachdem Sie die Klasse EDCLIB.VCX gewählt haben, sollte die Formular-Steuerelemente-Toolbar wie die in Abbildung 1 aussehen. Lassen Sie ein EDC-Objekt auf den Klassen-Designer fallen, und es erscheint wie in Abbildung 1. Wenn Sie das Symbol sehen, das die Anwesenheit der Klasse im DBC anzeigt, denken Sie daran, daß die EDC-Klasse kein visuelles Objekt auf der Toolbar ist. 

Abbildung 1: lsnToolBarEDC enthält ein eingebettetes EDC-Objekt und kann auf dessen Methoden wie auf die jedes anderen Kontrollelements zugreifen. 

Nun können Sie die Dienste des EDC von der Toolbar aufrufen. Wie Sie aus dem anderen Artikel noch wissen, benutzt die lsnToolbar eine eigene Methode SetState(), um zu entscheiden, welche Optionen zu den aktuellen Aktionen des Anwenders und der Umgebung passen. Listing A zeigt Ihnen den Code der Methode SetState() für die Klasse lsnToolbarEDC. 

Listing A 
* lsnToolBarEDC.SetState() 
THIS.SetAll("Enabled", NOT EMPTY(DBC())) 
IF EMPTY(DBC()) AND NOT EMPTY(THIS.cDatabase) 
THIS.cDatabase = "" 
THIS.oEDC.Close() 
ENDIF 

In diesem Fall deaktiviert SetState() die Buttons der Toolbar, wenn kein DBC als der aktuelle gewählt (oder wenn kein DBC offen) ist. SetState() prüft auch, ob die Toolbar vorher einen DBC registriert hat, indem es eine benutzerdefinierte cDatabase-Eigenschaft ausliest. 

An diesem Punkt fügen Sie die cProperty-Eigenschaft der lsnToolbarEDC-Eigenschaft hinzu und ergänzen die Toolbarklasse mit einer Methode OpenDBC(). Listing B zeigt Ihnen den Code für diese Methode. 

Listing B 
* lsnToolBarEDC.OpenEDC 
PRIVATE lreturn 
* nicht LOCAL, für den Fall, daß wir den Wert bei einem 
* einem Fehler benutzen wollen. Sie können auch eine 
* eigene Eigenschaft der Toolbar benutzen 
lreturn = .F. 
IF NOT EMPTY(THIS.cDatabase) AND ; 
THIS.cDatabase # SET("DATABASE") 
THIS.oEDC.Close() 
ENDIF

DO CASE 
CASE THIS.cDatabase == SET("DATABASE") 
* nichts zu tun 
lreturn = .T. 
CASE THIS.oEDC.lOpen("Master") 
* Benutzen Sie #DEFINE oder eine 
* andere Methode, um die Feldnamen 
* nicht wie hier hartcodiert festzuhalten, 
* wenn Sie wollen... 
* Ich kontrolliere hier nicht auf Fehler oder ob 
* Änderungen im DBC vorgenommen wurden, 
* da ich hier nur das Prinzip darstellen will. 
THIS.cDatabase = SET("DATABASE") 
lreturn = .T. 
OTHERWISE 
WAIT WINDOW "Zugriff auf EDC nicht möglich!" 
ENDCASE

RETURN lreturn 

Nun können Sie einen Button erstellen, mit dem Sie auf die erforderlichen Funktionen des EDC zugreifen können. In meinem Beispiel benutzt cmdSetInputMask den Code in Listing C, um die Eigenschaften zu setzen, die wir in der EDC-Tabelle bereits vorgesehen haben. 

Listing C 
* lsnToolBarEDC.cmdSetInputMask.Click() 
IF NOT THIS.Parent.OpenEDC() 
RETURN 
ENDIF 

IF EMPTY(ALIAS()) 
USE ? AGAIN 
ENDIF 

IF EMPTY(ALIAS()) OR NOT INDBC(ALIAS(),"TABLE") 
= MESSAGEBOX("You need to SELECT a table in " + ; 
the current DBC!" + CHR(13)+CHR(13) + "EDC " + ; 
"Toolbar works only on entities in the current DBC.", ; 
16,""Choose from current DBC") 
RETURN 
ENDIF 

LOCAL oDialog 
IF NOT "LSNTOOLS" $ UPPER(SET("CLASSLIB")) 
SET CLASSLIB TO lsnTools ADDITIVE 
ENDIF 

oDialog = CREATEOBJECT( ; 
"frmEDCFieldPropertySetter", ; 
"MyInputMask", THIS.Parent) 
THIS.Parent.Hide() 
oDialog.Show(1) 
THIS.Parent.Show() 

RETURN 

Wie Sie sehen können, ruft dieser Button eine modale Dialogbox auf, um die Eigenschaften vom Anwender abzufragen, genau wie der Datenbank-Designer den Tabellen-Designer aufruft, um die Feld- und Tabelleneigenschaften auszufüllen. 

Eine MODIFY STRUCTURE-Dialogbox im EDC-Stil 

Mein Dialogbox-Beispiel EDCFieldPropertySetter ist Teil von LSNTOOLS.VCX. Wie Abbildung 2 zeigt, stellt sie uns eine Dropdown-Liste der Felder der aktuellen Tabelle zur Verfügung und erlaubt es uns, eine Eigenschaft für jedes Feld zu setzen. (In Abbildung 2 sehen Sie die lsnToolbarEDC nicht, da die Klick-Methode des SetInputMask-Button die Toolbar verbirgt, bevor sie die Dialogbox anzeigt.) 

Da sie den Namen der erweiterten Eigenschaft eingeben, die Sie mit frmFieldPropertySetter setzen wollen, können Sie ihn _ oder eine von ihm abgeleitete Klasse _ für viele verschiedene Buttons verwenden, zum Beispiel die cmdSetInputMask. Weil der Toolbar-Button mit THIS.Parent auch eine Toolbar-Referenz an die Dialogbox schickt, hat die Dialogbox Zugriff auf das in EDC eingebettete Objekt und kann seine Methoden aufrufen. Listing D zeigt Ihnen, wie das Init die frmEDCFieldPropertySetter-Dialogbox den erweiterten Feldnamen und das Owner-Objekt der Toolbar als Parameter empfängt und anschließend diese Informationen benutzt. 

Listing D 
LPARAMETERS tcPropertyToSet, toToolbarOwner 

* Sie sollten hier die korrekte Parameterübergabe 
* mit ASSERT prüfen. 

THIS.autocenter = .T. 
=aFields(THIS.aFieldList) 
THIS.cboFields.Requery() 
THIS.cExtendedProperty = tcPropertyToSet 
THIS.oToolbar = toToolbarOwner 
THIS.cboFields.Valid() 
THIS.Caption="Specify "+THIS.cExtendedProperty + ; 
" for "+PROPER(ALIAS()) 
THIS.cmdStoreProperty.Caption = "Confirm " + ; 
THIS.cExtendedProperty+" Change" 

Wenn der Entwickler einen neuen Wert für MyInputMask bestätigt, speichert der Confirm-Button diese Änderung im EDC ohne eine explizite Referenz auf MyInputMask oder das Feld, für das die Eigenschaft gesetzt wurde. Die Click-Methode benutzt den folgenden Code: 

THISFORM.oToolbar.oEDC.uSetProp( ALIAS()+":"+ ; 
THISFORM.aFieldList[ ; 
THISFORM.cboFields.Value, 1], "FIELDS", ; 
THISFORM.cExtendedProperty, 
THISFORM.txtProperty.Value) 

Ein getrennter Button dieser Toolbar sollte in der gleichen Dialogbox die HelpContextID setzen. Außerdem sollten andere Buttons andere Dienste des EDC unter Verwendung verschiedener Mechanismen hinzufügen. Zum Beispiel könnten Sie zusätzliche EDCs öffnen oder in dem bestehenden EDC zusätzliche erweiterte Eigenschaften zuordnen, so daß Sie möglicherweise keinen Code wie KICKSTRT.PRG benötigen, um neue EDC-Features zu initialisieren. In meiner Version der lsnToolbarEDC habe ich einen Dummy-Button erstellt, um diesen Fakt aufzuzeigen, wie Sie in Abbildung 3 sehen können. Viele von Ihnen setzen mit eigenen Buildern im Formular- und Klassendesigner eine ganze Reihe Eigenschaften für die Objekte. Ich könnte mir vorstellen, daß Sie Ihre eigenen EDC-Dialogboxen herstellen, um alle erweiterten Eigenschaften für ein Feld oder alle Felder einer Tabelle gemeinsam zu setzen. 

Abbildung 2: Einstellung von zusätzlichen selbstdefinierten Eigenschaften im DBC über eine eigene Maske. 

Die Toolbar verfügbar machen 

Wie Sie noch aus dem Artikel „Dynamische Toolbars für dynamische Produktivität" wissen, machen Sie die lsnToolbar mit einem einfachen CREATEOBJECT() für Ihre Entwicklungsumgebung verfügbar. Sie könnten es aber auch in Ihrem VFPSTART-Programm aufrufen oder als eine Option in Ihr Menü aufnehmen. Wie Sie im Programm DBDESIGN.PRG in Listing E sehen können, sollten Sie vorsichtig sein, wenn Sie den Datenbank-Container öffnen, um ihn sowohl für ein direktes Editieren als auch für die Verwendung mit EDC zu benutzen. 

Listing E 
* DBDESIGN.PRG Beispiel lsnToolBarEDC 

IF TYPE("_tbrEDC") # "O" OR ISNULL(_tbrEDC) 
SET CLASSLIB TO lsnTools ADDITIVE 
PUBLIC _tbrEDC 
_tbrEDC = createobject("lsnToolbarEDC") 
_tbrEDC.Show() 
ENDIF

LOCAL laTemp[1,2] 
IF ADATABASES(laTemp) > 0 AND ; 
ASCAN(laTemp, "TEST") > 0 
SET DATA TO Test 
CLOSE DATA 
ENDIF 
OPEN DATA Test EXCLUSIVE 
MODI DATA Test NOWAIT 
RETURN 

Wahrscheinlich wollen Sie in Ihrer Version von DBDESIGN nicht den Namen einer speziellen Datenbank festschreiben. Das Objekt lsnToolbarEDC bleibt inaktiv, bis Sie einen DBC öffnen, aber Sie können es zu jeder Zeit instantiieren. Seine SetState()-Methode, die von dem Timer benutzt wird, der mit dem Objekt zusammenarbeitet, deaktiviert diese Funktionen, bis ein DBC verfügbar ist. Sobald ein DBC geöffnet wird, wird ihm von der Toolbar automatisch die entsprechende EDC-Tabelle zugeordnet. 

Die letzten Stücke des Puzzles 

Sie wissen nun, wie Sie einen EDC erstellen und wie Sie ihm ein visuelles Front-End während des Designs hinzufügen. Aber wie bekommen Sie die EDC-Erweiterungen in den Prozeß der Dateneingabe oder einen anderen Prozeß Ihrer Applikation? 

Wir müssen zwei Teile dieser Aufgabe unterscheiden _ Design und Runtime. Sie benötigen EDC-aktivierte Kontrollelemente in Ihren Formularen, und Sie müssen Ihren Formularen und anderen Geschäftsobjekten ein EDC-Objekt hinzufügen, wie Sie es bereits in der Toolbar getan haben. Dann, zur Laufzeit, können die Kontrollelemente die uGetProp()-Methode Ihres EDC-Objekts aufrufen, so wie Sie es während des Designs gesehen haben, als diese Methode von den Buttons der Toolbar und der modalen Dialogbox aufgerufen wurde. 

Abbildung 3: lsnToolBarEDC bietet durch den frmEDCFieldPropertySetter viele Dienste an. Mit zusätzlichen Dialogboxen dient die Toolbar als „Datenbank- und EDC-Builder". 

Mit einem „EDC-aktivierten" Kontrollelement meine ich eines, das Ihre EDC-Eigenschaften den Eigenschaften der Kontrollelemente zuordnet. Ich habe LSNTOOLS.VCX die Klasse txtEDCMask hinzugefügt, um Ihnen zu zeigen, was ich damit meine. Die Init()-Methode enthält den folgenden Code: 

THIS.InputMask = NVL( THISFORM.oEDC.uGetProp( ; 
THIS.ControlSource, „Field", „MyInputMask"), „") 

Ich benutze die Funktion NVL() für den Fall, daß das oEDC-Objekt .NULL. zurückgibt und damit anzeigt, daß dieses Feld keine erweiterte Eigenschaft MyInputMask registriert hat. In diesem Fall initialisiere ich die InputMask-Eigenschaft des Kontrollelements mit einem Nullstring. 

Wenn Sie es bevorzugen, keine EDC-aktivivierten Klassen zu erstellen, können Sie trotzdem mit EDC-aktivierten Kontrollelementen arbeiten, indem Sie mit einem Builder diesen den Code der Init()-Methode ihres Kontrollelements hinzufügen _ der Effekt des Codes während der Laufzeit ist identisch. Die Benutzung der Klassen ist zwar bessere OOP, aber die Verwaltung der Lösung mit dem Builder erlaubt es Ihnen, Felder per Drag-and-Drop direkt aus der Datenumgebung oder dem Projektmanager auf Ihr Formular zu ziehen und dabei Kontrollelemente zu erzeugen, die automatisch an die Daten gebunden sind. Da diese Kontrollelemente direkt von den Basisklassen von VFP abgeleitet sind, müssen Sie einen Builder einsetzen, um den zusätzlichen Init()-Code hinzuzufügen. 

Ich habe TEST.SCX mit mehreren dieser EDC-aktivierten Felder erstellt, um Ihnen das Endresultat zu zeigen. Das Formular vollführt einige phantasievolle Schritte, um den Kontrollelementen zu erlauben, diese Tabelle in TEST.DBC zu binden, ohne die einzelnen Kontrollelemente zu editieren. Der Code von TEST.SCX ist nicht weiter wichtig. Der einzige signifikante Code befindet sich im eingebetteten EDC-Objekt, das wir auf das Formular fallen ließen. Es beinhaltet die gleiche Technik, die wir bereits bei der Toolbar benutzt haben. 

Die Init()-Methode des EDC-Objekts setzt die aktuelle DATABASE TO Test und ruft dann die Funktion lOpen() auf, um den passenden EDC zu öffnen. Das Resultat sehen Sie in Abbildung 4. 

Abbildung 4: EDC-aktivierte Kontrollelemente basieren auf einem unsichtbaren EDC-Objekt des Formulars 

Jetzt machen Sie's selber aber nicht zu schnell 

In diesem Artikel sind wir mit einer berauschenden Geschwindigkeit durch einige neue Techniken gehetzt. Nun brauchen Sie erst einmal etwas Zeit, das alles nachzuvollziehen und richtig einzuschätzen. 

Stellen Sie sicher, daß Sie die Tools für die Erweiterung von VFP besitzen. Nehmen Sie sich Zeit, die VFP-Features darauf zu prüfen, ob sie Verbesserungen benötigen. Nehmen Sie einen Tip von Tom Rettig an: Designen Sie Ihre Verbesserungen so, daß sie sich perfekt in die existierenden Features einfügen. Genauso wie die EDC-Methoden die eingebauten DBC-Zugriffsfunktionen nachgebildet haben und durch die getrennten Tabellen der EDCs den DBC anderen Tools sauber hinterlassen, müssen Sie sicherstellen, daß Ihre Erweiterungen die Umgebung nicht „übernehmen". 

Das Erstellen von Entwicklertools erfordert Geduld. Auf jeden Fall kann das Erstellen von Tools ein gutes Training für jede objektorientierte Entwicklung sein, die Sie durchführen. 

Lisa Slater Nicholls ist von Neuseeland aus international in den Bereichen Training und Beratung tätig. Ihr Schwerpunkt ist der Review von Programmcode, Performance und Benutzeroberflächen von in Entwicklung befindlichen Applikationen unter Visual FoxPro. Sie können Sie per eMail unter lisa@softspoken.co.nz erreichen. 

